**************************************************************

*-- VFPsql

*-- parameters :tcDatabase,tcServer,tcUserId,tcPassword

*-- sql Connect

**************************************************************

FUNCTION VFPsql()

PARAMETERS tcDatabase,tcServer,tcUserId,tcPassword

SET MULTILOCKS ON 

LOCAL lcsqlConnectStr

PUBLIC gcsqlConnectstr

IF TYPE("gcsqlConnectstr")<>"C"

gcsqlConnectstr = ""

ENDIF 

lcsqlConnectStr =''

IF TYPE("_gnsqlConnectHandle") = "N" AND _gnSqlConnectHandle > 0

disVFPsql()

ENDIF 

PUBLIC _gnsqlConnectHandle

_gnsqlConnectHandle =0

IF PARAMETERS()>=3 AND VARTYPE(tcDatabase)='C' AND VARTYPE(tcServer)='C' AND VARTYPE(tcUserId)='C'

IF EMPTY(tcPassword)

tcPassword = ''

ENDIF 

lcsqlConnectStr ="driver={Sql Server};Database=&tcDatabase;server=&tcServer;uid=&tcUserId;pwd=&tcPassword"

_gnsqlConnectHandle = SQLSTRINGCONNECT(lcSqlConnectStr)

ELSE

IF gcsqlConnectstr <> ""

lcsqlConnectStr = gcSqlConnectstr

_gnsqlConnectHandle = SQLSTRINGCONNECT(lcSqlConnectStr)

ENDIF 

ENDIF 

gcsqlConnectstr = lcSqlConnectStr

LOCAL iLoopCount

iLoopCount = 1

DO WHILE _gnsqlConnectHandle <= 0 AND iLoopCount <=3

DO prgDataConnect

iLoopCount = iLoopCount + 1

ENDDO 

RETURN _gnsqlConnectHandle>0

*- VFPsql END

************************************************

*-- disVFPsql()

*-- sql Disconnect

************************************************

FUNCTION disVFPsql()

IF TYPE("_gnsqlConnectHandle")<>'N' OR _gnSqlConnectHandle<=0

Return

ENDIF

sqlDISCONNECT(_gnSqlConnectHandle)

RELEASE _gnsqlConnectHandle

RETURN

*- DisVFPsql END

***************************************

*-- sqlDo()

*-- PARAMETERS tcsqlString: 查询语句

*--tcCurName : 结果集存放表

*- 功能： 执行查询

**************************************

FUNCTION sqlDo

LPARAMETERS tcsqlString,tcCurName

LOCAL lnReturn

*-- 检查连接

IF TYPE("_gnsqlConnectHandle")<>"N" or _gnSqlConnectHandle<=0

IF !VFPsql()

*-- 连接失败，返回 0

RETURN 0

ENDIF 

ENDIF

IF TYPE('tcsqlString')<>'C' OR EMPTY(tcSqlString)

RETURN 0

ENDIF

IF EMPTY(tcCurName) OR TYPE("tcCurName") <>'C'

lnReturn = sqlEXEC(_gnSqlConnectHandle,tcSqlString)

ELSE

lnReturn = sqlEXEC(_gnSqlConnectHandle,tcSqlString,tcCurName)

ENDIF

RETURN lnReturn

*- sqlDo END

*-- Opendb

********************************************************

*-- OpenDB

*

*- 功能：打开table Or View

*

*-参数说明 tcCurName : 临时表名

*- tcsqlTblName: Sql数据表名(默认tcCurName)

* tcFldsList : 字段列表(默认为全部 *)

**********************************************************

FUNCTION OpenDB

LPARAMETERS tcCurName,tcsqlTblName,tcFldsList

IF VARTYPE(tcCurName)<>"C"

MESSAGEBOX("参数错误,不能打开数据表或视图!",48,"错误")

RETURN .F.

ENDIF 

IF EMPTY(tcsqlTblName)

tcsqlTblName = tcCurName

ENDIF 

IF EMPTY(tcFldsList)

tcFldsList = "*"

ENDIF

*-- 执行查询

lnReturn = sqldo("select &tcFldsList From &tcSqlTblName ","&tcCurName.")

*-- 打开开放式表缓冲

IF lnReturn > 0

cursorsetprop("buffering",5, "&tcCurName.")

ENDIF 

RETURN lnReturn>0

*-- OpenDB END

********************************************************

*-- CloseDB

*

*- 功能：关闭table Or View

*

*-参数说明 tcCurName : 临时表名

*-

**********************************************************

FUNCTION CloseDB

LPARAMETERS tcCurName

IF VARTYPE(tcCurName)<>"C"

MESSAGEBOX("参数错误,非法数据表或视图!",48,"错误")

RETURN .F.

ENDIF 

IF SELECT(tcCurName) > 0

USE IN &tcCurName

ENDIF

RETURN .T.

*-- CloseDB END

**********************************************************************

*- SaveData

*-- 

*-- 参数说明 tcCurName : 临时表名

*-- tcsqlTblName: Sql数据表名(默认tcCurName)

*-- tcKeyFld : 关键字段名称（一定要有，否则默认为ID）,

*-- 程序要根据该字段来定位记录，

*-- 功能：从临时表保存到 sql数据库中，注意，临时表要启用缓冲模式

**********************************************************************

FUNCTION savedata

LPARAMETERS tcCurName,tcsqlTblName,tcKeyFld

LOCAL lcFldList,lcFldListTmp,lnNextModifyRec,lcFldValue,lnFldCnt

LOCAL lcsqlString,lnSqlReturn

LOCAL lcOldAlias,lcOldDele,lcOldSafe

LOCAL laFlds(1)

LOCAL i,llNeedAddDH

IF TYPE("tcCurName")<>'C' OR ;

EMPTY(tcCurName) OR ;

SELECT(tcCurName)=0

MESSAGEBOX("表名错误！",48,"错误提示")

RETURN

ENDIF

lcOldDele = SET("Deleted")

lcOldSafe = SET("Safety")

SET DELETED OFF 

SET SAFETY OFF 

lcOldAlias = ALIAS()

SELECT (tcCurName)

IF cursorgetprop('buffering')=1

MESSAGEBOX("数据表没有启用缓冲模式，程序无法识别！",48,"错误提示")

RETURN .F.

ENDIF

IF EMPTY(tcsqlTblName) OR TYPE("tcSqlTblName")<>'C'

tcsqlTblName = tcCurName

ENDIF

IF EMPTY(tcKeyFld)

tcKeyFld = 'Id'

ENDIF 

lnFldCnt = AFIELDS(laFlds)

lcFldList = ''

FOR i = 1 TO lnFldCnt

IF !EMPTY(lcFldList)

lcFldList = lcFldList +','

ENDIF

lcFldList = lcFldList + laFlds(i,1)

ENDFOR

lnNextModifyRec = GETNEXTMODIFIED(0)

lnsqlReturn = SQLDO("BEGIN TRANSACTION")

IF lnsqlReturn =-1

RETURN .F.

ENDIF

DO WHILE lnNextModifyRec <>0

GO lnNextModifyRec

lcFldListtmp = ''

lcsqlString = ''

DO case

CASE DELETED()&&删除

lcsqlString = "Delete From "+tcSqlTblName + IIF(TYPE(tcKeyFld) $ "NI",;

" Where &tcKeyFld. = "+TRANSFORM(EVALUATE(tcCurName+".&tcKeyFld.")),;

" Where &tcKeyFld. = '"+TRANSFORM(EVALUATE(tcCurName+".&tcKeyFld."))+"'")

CASE RECNO()<0 &&新增

lcsqlString = "" &&"insert into "+tcSqlTblName +" ("+lcFldList+") Values ("

FOR i = 1 TO lnFldCnt

lcFldValue = EVALUATE(tcCurName+'.'+laFlds(i,1))

IF EMPTY(lcFldValue)

Loop

ENDIF

IF !EMPTY(lcFldListtmp)

lcFldListtmp = lcFldListtmp +','

ENDIF

lcFldListtmp = lcFldListtmp + laFlds(i,1)

DO case

CASE laFlds(i,2)$'NI'

lcFldValue = TRANSFORM(lcFldValue)

CASE laFlds(i,2)='L'

IF lcFldValue

lcfldValue = '1'

ELSE

lcfldValue = '0'

ENDIF

OTHERWISE

lcFldValue ="'"+ TRANSFORM(lcFldValue)+"'"

ENDCASE

IF !EMPTY(lcsqlString)

lcsqlString =lcSqlString +','

ENDIF

lcsqlString =lcSqlString + lcFldValue

ENDFOR

lcsqlString ="insert into "+tcSqlTblName +" ("+lcFldListtmp+") Values ("+lcSqlString +')'

OTHERWISE

lcsqlString = "Update "+tcSqlTblName +" set "

llNeedAddDH = .F.

FOR i = 1 TO lnFldCnt

IF GETFLDSTATE(i)<>2

LOOP

ENDIF

DO case

CASE laFlds(i,2)$'NI'

lcFldValue = laflds(i,1)+'='+TRANSFORM(EVALUATE(tcCurName+'.'+laFlds(i,1)))

CASE laFlds(i,2)='L'

IF EVALUATE(tcCursorName+'.'+laFlds(i,1))

lcfldValue = laflds(i,1)+'=1'

ELSE

lcfldValue = laflds(i,1)+'=0'

ENDIF

OTHERWISE

lcFldValue =laflds(i,1)+" = '"+ TRANSFORM(EVALUATE(tcCurName+'.'+laFlds(i,1)))+"'"

ENDCASE

IF NOT llNeedAddDH 

llNeedAddDH = .T.

Else

lcsqlString =lcSqlString +','

ENDIF

lcsqlString =lcSqlString + lcFldValue

ENDFOR

lcsqlString =lcSqlString + IIF(TYPE(tcKeyFld) $ "NI",;

" Where &tcKeyFld. = "+TRANSFORM(EVALUATE(tcCurName+".&tcKeyFld.")),;

" Where &tcKeyFld. = '"+TRANSFORM(EVALUATE(tcCurName+".&tcKeyFld."))+"'")

ENDCASE

lnsqlReturn = sqlDo(lcSqlString)

IF lnsqlReturn=-1

sqlDO("Rollback")

MESSAGEBOX("保存失败!"+CHR(13)+CHR(10)+lcsqlString)

_cliptext = lcsqlstring

TABLEREVERT(.T.)

RETURN .F.

ENDIF

lnNextModifyRec = GETNEXTMODIFIED(lnNextModifyRec)

ENDDO

lnsqlReturn = SQLDO(" COMMIT TRANSACTION ")

TABLEUPDATE(.T.)

IF !EMPTY(lcOldAlias) AND ALIAS()<>lcOldAlias

SELECT (lcOldAlias)

ENDIF

IF lcOldDele='ON'

SET DELETED on

ENDIF

IF lcOldSafe = 'ON'

SET SAFETY on

ENDIF

RETURN .T.

****************

*-- 调用sql设置

PROCEDURE prgdataconnect

LOCAL lcDataPath,lcTextString

LOCAL loForm

_gnsqlConnectHandle = 0

IF _gnsqlConnectHandle <=0

loFormCon = CREATEOBJECT("fmConnect")

loFormCon.Show(1)

ENDIF

**********************************************

*- 定义连接表单

*********************************************

DEFINE CLASS fmConnect as Form

&&-"driver={sql Server};Database=Hjasset;server=jianqy;uid=sa;pwd=")

Caption = 'sql服务器连接'

Width = 300

Height = 182

BorderStyle = 2

maxbutton = .F.

minButton = .F.

ShowWindow = 1

Autocenter = .T.

alwaysontop = .T.

desktop = .T.

windowType = 1

Name = 'fmConnect'

ADD OBJECT lblServer as label WITH Caption = '服务器：',;

left = 15,;

Width = 80,;

Top = 20,;

Height= 22,;

Alignment = 1,;

BackStyle = 0,;

Visible = .T.

ADD OBJECT txtServer as textbox WITH left = 95,;

Width = 160,;

Top = 17,;

Height= 25,;

Value = '(Local)',;

Visible = .T.

ADD OBJECT lblDatabase as label WITH Caption = '数据库：',;

left = 15,;

Width = 80,;

Top = 50,;

Height= 22,;

Alignment = 1,;

BackStyle = 0,;

Visible = .T.

ADD OBJECT txtDatabase as textbox WITH left = 95,;

Width = 160,;

Top = 47,;

Height= 22,;

Value = 'pubs',;

Visible = .T.

ADD OBJECT lblUid as label WITH Caption = '用户名：',;

left = 15,;

Width = 80,;

Top = 80,;

Height= 22,;

Alignment = 1,;

BackStyle = 0,;

Visible = .T.

ADD OBJECT txtUid as textbox WITH left = 95,;

Width = 160,;

Top = 77,;

Height= 22,;

Value = 'sa',;

Visible = .T.

ADD OBJECT lblPwd as label WITH Caption = '密 码：',;

left = 15,;

Width = 80,;

Top = 110,;

Height= 22,;

Alignment = 1,;

BackStyle = 0,;

Visible = .T.

ADD OBJECT txtPwd as textbox WITH left = 95,;

Width = 160,;

Top = 107,;

Height= 22,;

passwordchar= '*',;

Value = '',;

Visible = .T.

ADD OBJECT btnConnect as Commandbutton WITH left = 35,;

Width = 100,;

Top = 150,;

Height= 22,;

Value = '',;

Visible = .T.,;

Caption = '连 接'

ADD OBJECT btnCancel as Commandbutton WITH left = 165,;

Width = 100,;

Top = 150,;

Height= 22,;

Value = '',;

Visible = .T.,;

Caption = '取 消'

ADD OBJECT cntShp1 as Container WITH left = 0,;

Width = 300,;

Top = 142,;

Height= 1,;

Visible = .T.,;

SpecialEffect = 1 

PROCEDURE destroy

PROCEDURE load

PROCEDURE init

PROCEDURE txtPwd.Keypress

LPARAMETERS nKeyCode, nShiftAltCtrl

IF nKeyCode = 3 OR nKeyCode = 24

NODEFAULT

ENDIF 

PROCEDURE btnConnect.click

LOCAL lcServer,lcDatabase,lcUid,lcPwd,lcsqlStrConnect

lcServer = ALLTRIM(thisform.txtServer.Value)

lcDatabase = ALLTRIM(thisform.txtDatabase.Value)

lcUid = ALLTRIM(thisform.txtUid.Value)

lcPwd = ALLTRIM(thisform.txtPwd.Value)

IF EMPTY(lcServer)

MESSAGEBOX("请输入sql服务器的名称!",48,"系统提示")

thisform.txtServer.setfocus

RETURN

ENDIF

lcsqlStrConnect="driver={Sql Server};"+;

IIF(EMPTY(lcDatabase),'',"Database="+lcDatabase+";")+;

"server="+lcServer+";uid="+lcUid+";pwd="+lcPwd

_gnsqlConnectHandle = SQLSTRINGCONNECT(lcSqlStrConnect)

gcsqlConnectstr = lcSqlStrConnect

ON ERROR 

IF _gnsqlConnectHandle > 0

MESSAGEBOX("连接成功!",64,"sql连接")

thisform.Release 

ELSE

MESSAGEBOX("连接失败!"+CHR(13)+CHR(10)+"请重新输入连接项",64,"sql连接")

thisform.txtServer.setfocus

ENDIF

PROCEDURE btnCancel.click

thisform.Release 

ENDDEFINE

*************

*-- 以pubs 为例

&&连接sql

VFPsql()

&&打开authors

opendb("authors")

&&修改

repl au_lname with 'dfjkdjfkdj' for au_id = '722-51-5454'

&&保存到sql

savedata("authors","authors","au_id")

*************

一、采用ado方式,可能代码会很多,但是会很灵活,而且可以让前台没有dbc文件了，如果采用远程视图方式,就不会写这么多代码了,但是必须要定义DBC（数据库）用于保存远程视图及连接等，前一种方式，楼主已经写的比较通俗简明了，我对后一种简单补充一下。

1.建立数据库，然后建立与后台数据库的连接(当然可以用INI文件等先保存定义，然后动态设定）

2.建立远程视图表

3.打开表同时设置表的缓冲级别,一般为行缓冲和表缓冲

4.修改后如果保存=tableupdate(.t.),放弃=tablerevert(.t.)

5.可以通过捕获返回的错误信息,显示给用户.如

IF TABLEUPDATE(.T.)

=MESSAGEBOX("更新完毕!",48,"系统信息")

ELSE

=AERROR(ERR)

IF !EMPTY(ERR(2))

=SHOWWARN("错误信息:"+ALLTRIM(SUBSTR(UPPER(ERR(2)),AT('[sql SERVER]',UPPER(ERR(2)),1)+12)))

ENDIF

XX=SHOWCHOICE("是否取消此次操作?")

IF XX=1

=TABLEREVERT(.T.)

ELSE

RETURN

ENDIF

ENDIF

=REQUERY()

二、当然了，现在最好使用的还是CursorAdapter方式，它即有了ado方式的灵活，又有了远程视图的简易，还有了后台升级方便等优点。具体使用请参考网上的一些说明，在些从简了。

VFP中用spt访问sqlServer数据库

新一篇: 批量更改sqlserver数据库所有者的对象

VFP因它简单易学，可快速建立应用软件而深受广大程序员喜爱，但其数据库系统不安全也令广大用户非常头痛。随着MSsqlServer数据库系统的推广应用，其强大的安全性能普遍受到好评。笔者在长期的编程实践中，发现利用vfp的SQLpass-through(spt)技术结合MSSQLServer数据库系统，也能像VB、Delphi、PowerBuilder一样轻松开发出优秀的客户/服务器(C/S)应用软件。现通过编写一个简单的通信录例子和广大VFP爱好者共同探讨。

　　一、服务器端MSsqlServer数据库设计

1.在MSsqlServer中建立名为“SFXT”的数据库。

2.在SFXT数据库中建立名为“通信录”的数据表，表结构如下：

列名（字段名）　数据类型 　长度

姓名chr8

生日datetime8

电话chr11

emailchr30

并将姓名设置为主键。

3.在“SFXT”数据库中建立查询出所有记录的存储过程。SP_SearchAll

CREATEPROCEDURE[SP_SearchAll]

AS

select*from通信录orderby姓名

return

4.在“SFXT”数据库中建立插入新记录的存储过程SP_InsertData。

CREATEPROCEDURE[SP_InsertData]

@name[char](10)，

@birthday[datetime]，

@telephone[char](11)，

@email[char](30)

AS

insertinto通信录(姓名，生日，电话，email)values(@name，@birthday，@telephone，@email)

return

二、客户端VFP的sqlpass-through技术设计

1.建立如(^15020603b^)的VFP表单界面。

2.建立名为“SFXT”的ODBC数据源连接MSsqlServer中的SFXT数据库。

可通过执行上面VFP表单的按钮“建立OBDC数据源”来完成；或者通过运行Windows控制面板中的“OBDC数据源”来完成。主要设置包括选择sqlServer驱动程序，通信协议，登录标识与密码等。

“建立OBDC数据源”按钮.CLICK事件：

*函数说明sqlSTRINGCONNECT([cConnectString])

*省略连接字符串cConnectString时显示""sql数据源""对话框，可选择或新建数据源

sqlstringconnect()

3.两种连接MSsqlServer数据源的方式。

使用现有数据源名称建立连接，“建立连接方式1”按钮.CLICK事件。

publicvodbc,vuser,vpwd,vconn

vodbc='sfxt' &&连接sqlServer数据库ODBC数据源名称

vuser='sa' &&访问sqlServer数据库的登录用户名，sa为系统用户

vpwd='5213' &&用户登录密码，为sa系统用户设置的密码

vconn=sqlCONNECT(vodbc,vuser,vpwd)

ifvconn>0

messagebox('连接成功！',,'ODBC数据源')

else

messagebox('连接失败！',,'ODBC数据源')

endif

使用连接字符串建立数据源连接，“建立连接方式2”按钮.CLICK事件。

publicvconn

vconn=sqlSTRINGCONNECT('dsn=sfxt:uid=sa:pwd=5213')

ifvconn>0

messagebox('连接成功！'，，'ODBC数据源')

else

messagebox('连接失败！'，，'ODBC数据源')

endif

4.数据源连接的主要参数设置，“活动连接属性设置”按钮.CLICK事件。

*注：用函数sqlgetprop()可返回设置的参数值

sqlsetprop(vconn，""asynchronous""，.f.)&&取假值时为结果集同步返回：取真值时为异步返回

sqlsetprop(vconn,""ConnectTimeOut"",15)&&连接超时等待秒数设置，可取值0至600

sqlsetprop(vconn,""IdleTimeout"",0) &&空闲超时间隔秒数，取值0为无限期超时等待

sqlsetprop(vconn,""Transactions"",1) &&取值1时为自动处理远程事务：取值2时为手工处理

5.“执行存储过程查询数据”按钮.CLICK事件。

*函数说明sqlEXEC(nConnectionHandle,[cSQLCommand，[CursorName]])

*nConnectionHandle 当前数据源活动连接句柄

*csqlCommand 执行SQLServer存储过程的SQL语句表达式

*CursorName 返回执行结果临时表的名称

sqlexec(vconn,""executeSP_SearchAll"",""我的通信录"")

browse

6.“执行sql语句查询数据”按钮.CLICK事件。

*函数说明sqlEXEC(nConnectionHandle,[cSQLCommand,[CursorName]])

*nConnectionHandle 当前数据源活动连接句柄

*csqlCommand 需发送SQL语句表达式

*CursorName 返回执行结果临时表的名称

sqlEXEC(vconn,'SELECT*FROM通信录','我的通信录')

browse

7.“向存储过程传替参数插入新记录”按钮.CLICK事件。

localvname,vbirthday,vtelephone,vemail,vsql

*随机产生新记录举例

vname='姓名'+sys(3) &&姓名

vbirthday=dtoc(date()-int(rand()*10000)) &&生日

vtelephone=sys(3) &&电话

vemail=sys(3)+'@hotmail' &&电子邮箱

*将传递到存储过程的参数转换成字符串，并加引号形成sql语句。

vsql=""executeSP_InsertData""+""'""+vname+""','""+vbirthday+""','""+vtelephone+""','""+vemail+""'""

ifsqlexec(vconn，vsql)>0 &&发送SQL语句

messagebox('插入新记录成功！'，，'信息')

else

messagebox('插入新记录不成功！'，，'信息')

endi

sqlexec(vconn，""executeSP_SearchAll""，""我的通信录"")

browse

8.“设置当前表的属性修改数据”按钮.CLICK事件。

*注：用函数cursorgetprop()可返回设置的参数值

cursorsetprop('BatchUpdateCount',100) &&发送到缓冲表的远程数据源的更新指令的数目

cursorsetprop('Buffering',3) &&设置当前表为开放式行缓冲

cursorsetprop('FetchSize',-1) &&从远程表中提取全部查询记录

cursorsetprop('KeyFieldList','姓名') &&指定远程表的主关键字段

cursorsetprop('SendUpdates',.t.) &&当前缓冲表更改内容时发送sql语句更新远程表

cursorsetprop('Tables','通信录') &&指定连接的远程表名

*本地缓冲表与远程表字段对应关系

cursorsetprop('UpdateNameList','姓名通信录.姓名,生日通信录.生日,电话通信录.电话,email通信录.email')

*指定可更新字段列表

cursorsetprop('UpdatableFieldList','姓名,生日,电话,email')

cursorsetprop('UpdateType',1) &&远程表更新方式，替换方式

cursorsetprop('WhereType',3) &&更新sql语句中where子句包含主关键字与被修改过的字段

browse &&修改缓冲表数据，移动记录指针后，自动发送sql语句更新远程表

9.“断开指向数据源的连接”按钮.CLICK事件。

sqldisconnect(vconn)

10.“关闭”按钮.CLICK事件。

thisform.release

以上我们利用VFP的sqlpass-through技术编写了一个简单的通信录管理程序，实现对SQLServer数据库中数据的插入、查询、修改等基本功能，是一个典型的客户/服务器(C/S)结构的应用程序。希望对读者今后开发软件有所帮助。

